
18 Imitation Learning

Previous chapters have assumed that a reward function is either known or that re-
wards are received while interacting with the environment. For some applications,
it may be easier for an expert to demonstrate the desired behavior rather than spec-
ifying a reward function. This chapter discusses algorithms for imitation learning,
where the desired behavior is learned from expert demonstration. We will cover
a variety of methods ranging from very simple likelihood-maximization methods
to more complicated iterative methods that involve reinforcement learning.1 1 Additional methods and applica-

tions are surveyed by A. Hussein,
M.M.Gaber, E. Elyan, andC. Jayne,
‘‘Imitation Learning,’’ ACM Com-
puting Surveys, vol. 50, no. 2, pp. 1–
35, 2017.

18.1 Behavioral Cloning

A simple form of imitation learning is to treat it as a supervised learning problem.
This method, called behavioral cloning,2 trains a stochastic policy πθ parameterized 2 D.A. Pomerleau, ‘‘Efficient Train-

ing of Artificial Neural Networks
for Autonomous Navigation,’’ Neu-
ral Computation, vol. 3, no. 1, pp. 88–
97, 1991.

by θ to maximize the likelihood of actions from a dataset D of expert state-action
pairs:

maximize
θ

∏
(s,a)∈D

πθ(a | s) (18.1)

As done in earlier chapters, we can transform the maximization over the product
over πθ(a | s) to a sum over log πθ(a | s).

Depending on howwe want to represent the conditional distribution πθ(a | s),
we may compute the maximum likelihood estimate of θ analytically. For example,
if we use a discrete conditional model (section 2.4), θ would consist of all of the
counts N(s, a) from D and πθ(a | s) = N(s, a)/ ∑a N(s, a). Example 18.1 applies
a discrete conditional model to data from the mountain car problem.

If we have a factored representation of our policy, we can use a Bayesian
network to represent the joint distribution over our state and action variables.
Figure 18.1 shows an example. We can learn both the structure (chapter 5) and

Consider using behavioral cloning on expert demonstrations for the moun-
tain car problem (appendix F.4). We are given 10 rollouts from an expert
policy. We fit a conditional distribution and plot the results. The continuous
trajectories were discretized with 10 bins each for position and for speed.

The state space is not fully covered by the exert demonstrations, which is
typical for imitation learning problems. The resulting policy may perform
well when used in regions with coverage, but it assigns a uniform distribution
to actions in regions without coverage. Even if we start in a region with
coverage, we may transition to regions without coverage due to stochasticity
in the environment.

Example 18.1. A demonstration
of behavioral cloning applied to
the mountain car problem. The
light blue regions are areaswithout
training data, resulting in poor pol-
icy performancewhen the agent en-
counters those states.

18.1 . behavioral cloning 349

parameters (chapter 4) from the data D. Given the current state, we can then
infer the distribution over actions using one of the inference algorithms discussed
earlier (chapter 3).

s1 s2 s3

a1 a2

P(s1) P(s2) P(s3)

P(a1 | s1, s2) P(a2 | s1, s2, s3)

Figure 18.1. Bayesian networks can
be used to represent a joint distribu-
tion over the state and action vari-
ables. We can apply an inference al-
gorithm to generate a distribution
over actions given the current val-
ues of the state variables.

We can use many other representations for πθ. For example, we might want
to use a neural network, where the input corresponds to the values of the state
variables and the output corresponds to parameters of a distribution over the
action space. If our representation is differentiable, which is the case with neural
networks, we can attempt to optimize equation (18.1) using gradient ascent. This
approach is implemented in algorithm 18.1.

struct BehavioralCloning

α # step size

k_max # number of iterations

∇logπ # log likelihood gradient

end

function optimize(M::BehavioralCloning, D, θ)

α, k_max, ∇logπ = M.α, M.k_max, M.∇logπ

for k in 1:k_max

∇ = mean(∇logπ(θ, a, s) for (s,a) in D)

θ += α*∇

end

return θ

end

Algorithm 18.1. A method for
learning a parameterized stochas-
tic policy from expert demonstra-
tions in the form of a set of state-
action tuples D. The policy param-
eterization vector θ is iteratively
improved by maximizing the log
likelihood of the actions given the
states. Behavioral cloning requires
a step size α, an iteration count
k_max, and a log likelihood gradi-
ent ∇logπ.

The closer the expert demonstrations are to optimal, the better the resulting
behavioral cloning policy will perform.3 However, behavioral cloning suffers 3 U. Syed and R. E. Schapire, ‘‘A

Reduction from Apprenticeship
Learning to Classification,’’ in Ad-
vances in Neural Information Process-
ing Systems (NIPS), 2010.

from cascading errors. As discussed in example 18.2, small inaccuracies compound
during a rollout and eventually lead to states that are poorly represented in the
training data, thereby leading to worse decisions, and ultimately to invalid or
unseen situations. Though behavioral cloning is attractive due to its simplicity, cas-
cading errors cause the method to perform poorly on many problems, especially
when policies must be used for long time horizons.

Consider applying behavioral cloning to train a policy for driving an au-
tonomous race car. A human race car driver provides expert demonstrations.
Being an expert, the driver never drifts onto the grass or too close to a railing.
A model trained with behavioral cloning would have no information to go
off of when near a railing or when drifting onto the grass, and would not
know how to recover.

Example 18.2. A brief example of
the generalization issue inherent to
behavioral cloning approaches.

18.2 Dataset Aggregation

One way to address the problem of cascading errors is to correct a trained policy
using additional expert input. Sequential interactive demonstration methods alter-
nate between collecting data from an expert in situations generated by a trained
policy and using this data to improve this policy.

One type of sequential interactive demonstration method is called dataset
aggregation (DAgger) (algorithm 18.2).4 It starts by training a stochastic policy 4 S. Ross, G. J. Gordon, and J.A.

Bagnell, ‘‘A Reduction of Imitation
Learning and Structured Predic-
tion to No-Regret Online Learn-
ing,’’ in International Conference on
Artificial Intelligence and Statistics
(AISTATS), vol. 15, 2011.

using behavioral cloning. The policy is then used to run several rollouts from an
initial state distribution b, which are then given to an expert to provide the correct
actions for each state. The new data is aggregated with the previous dataset and
a new policy is trained. Example 18.3 illustrates this process.

These interactive demonstrations iteratively builds a dataset covering the re-
gions of the state space that the agent is likely to encounter, based on previous
learning iterations. With each iteration, newly added examples compose a smaller
fraction of the dataset, thereby leading to smaller policy changes. While sequen-
tial interactive demonstration can work well in practice, it is not guaranteed to
converge.

18.3 Stochastic Mixing Iterative Learning

Sequential interactive methods can also iteratively build up a policy by stochas-
tically mixing in newly trained policies. One such method is stochastic mixing
iterative learning (SMILe) (algorithm 18.3).5 It uses behavioral cloning in every

5 S. Ross and J.A. Bagnell, ‘‘Ef-
ficient Reductions for Imitation
Learning,’’ in International Confer-
ence on Artificial Intelligence and
Statistics (AISTATS), 2010.

iteration but mixes the newly trained policy in with the previous ones.
We start with the expert policy, π(1) = πE.6 In each iteration, we execute

6 We do not have an explicit rep-
resentation of πE. Evaluating πE
requires interactively querying the
expert as done in the previous sec-
tion.the latest policy π(k) to generate a new dataset, querying the expert to provide

struct DatasetAggregation

𝒫 # problem with unknown reward function

bc # behavioral cloning struct

k_max # number of iterations

m # number of rollouts per iteration

d # rollout depth

b # initial state distribution

πE # expert

πθ # parameterized policy

end

function optimize(M::DatasetAggregation, D, θ)

𝒫, bc, k_max, m = M.𝒫, M.bc, M.k_max, M.m

d, b, πE, πθ = M.d, M.b, M.πE, M.πθ

θ = optimize(bc, D, θ)

for k in 2:k_max

for i in 1:m

s = rand(b)

for j in 1:d

push!(D, (s, πE(s)))

a = rand(πθ(θ, s))

s = rand(𝒫.T(s, a))

end

end

θ = optimize(bc, D, θ)

end

return θ

end

Algorithm 18.2. The DAgger
method of dataset aggregation for
learning a stochastic parameter-
ized policy from expert demonstra-
tions. This method takes an ini-
tial dataset of state-action tuples D,
a stochastic parameterized policy
πθ(θ, s), an MDP 𝒫 that defines
a transition function, and an ini-
tial state distribution b. Behavioral
cloning, algorithm 18.1, is used in
each iteration to improve the pol-
icy.

An expert policy πE labels tra-
jectories sampled from the latest
learned policy to augment the
dataset. The original paper gener-
ated trajectories by stochastically
mixing in the expert policy. This
implementation is thus the origi-
nal DAgger with an extreme mix-
ing value of zero.

In practice an expert policy
would not exist, and calls to this
policy would be replaced with
queries to a human expert.

352 chapter 18. imitation learning

Consider using dataset aggregation to train a policy on the mountain car
problem where the reward is not observed. We use an expert policy that
accelerates in the direction of travel. In this example we train a policy using
the features:

f(s) = [1[v > 0], 1[v < 0], x, x2, v, v2, xv]

where x and v are the position and speed of the car.

Trajectories begin in red and proceed to blue. In the first iteration, the
agent behaves randomly, unable to make progress toward the goal (x ≥
0.6). With additional iterations, the agent learns to mimic the expert policy
of accelerating in the direction of travel. This behavior is apparent in the
new trajectories, which spiral outward, and the policy, which assigns high
likelihood to a = 1 when v > 0 and a = −1 when v < 0.

Example 18.3. Dataset aggregation
applied to the mountain car prob-
lem, with iterations running from
top to bottom. Trajectories accumu-
late in the dataset over time. The be-
havior of the agent improves with
each iteration.

18.4. maximum margin inverse reinforcement learning 353

the correct actions. Behavioral cloning is applied only to this new dataset to
train a new component policy π̂(k). This component policy is mixed together with
component policies from the previous iterations to produce a new policy π(k+1).

The mixing of component policies to generate π(k+1) is governed by a mixing
scalar β ∈ (0, 1). The probability of acting according to the expert policy is (1− β)k,
and the probability of acting according to π̂(i) is β(1− β)i−1. This scheme assigns
more weight to older policies under the hypothesis that older policy components
were trained on the states more likely to be encountered. With each iteration, the
probability of acting according to the original expert policy decays to zero. The
mixing scalar is typically small such that the agent does not abandon the expert’s
policy too quickly. Example 18.4 demonstrates this approach on mountain car.

18.4 Maximum Margin Inverse Reinforcement Learning

In many application settings, we do not have an expert that can be interactively
queried, but instead have a batch of expert demonstration trajectories. We will
assume that the expert demonstration data D consists of m trajectories. Each
trajectory τ in D involves a rollout to depth d. In inverse reinforcement learning, we
assume that the expert is optimizing some unknown reward function. From D,
we attempt to derive that reward function. With that reward function, we can use
the methods discussed in prior chapters to derive an optimal policy.

There are different approaches to inverse reinforcement learning. We generally
need to define a parameterization of the reward function. A common assumption
is that this parameterization is linear with Rφ(s, a) = φ>β(s, a), where β(s, a)
is a feature vector andφ is a vector of weightings. In this section, we will focus
on an approach known as maximum margin inverse reinforcement learning,7 where 7 P. Abbeel and A.Y. Ng, ‘‘Appren-

ticeship Learning via Inverse Re-
inforcement Learning,’’ in Interna-
tional Conference on Machine Learn-
ing (ICML), 2004.

the features are assumed to be binary. Since optimal policies remain optimal with
positive scaling of the reward function, this method additionally constrains the
weight vector such that ‖φ‖2 ≤ 1. The expert data activates each binary feature
with different frequencies, perhaps pursuing some and avoiding others. This
approach attempts to learn this pattern of activation and trains an agent to mimic
these activation frequencies.

354 chapter 18. imitation learning

struct SMILe

𝒫 # problem with unknown reward

bc # Behavioral cloning struct

k_max # number of iterations

m # number of rollouts per iteration

d # rollout depth

b # initial state distribution

β # mixing scalar (e.g., d^-3)

πE # expert policy

πθ # parameterized policy

end

function optimize(M::SMILe, θ)

𝒫, bc, k_max, m = M.𝒫, M.bc, M.k_max, M.m

d, b, β, πE, πθ = M.d, M.b, M.β, M.πE, M.πθ

𝒜, T = 𝒫.𝒜, 𝒫.T

θs = []

π = s -> πE(s)

for k in 1:k_max

# execute latest π to get new dataset D

D = []

for i in 1:m

s = rand(b)

for j in 1:d

push!(D, (s, πE(s)))

a = π(s)

s = rand(T(s, a))

end

end

# train new policy classifier

θ = optimize(bc, D, θ)

push!(θs, θ)

# compute a new policy mixture

Pπ = Categorical(normalize([(1-β)^(i-1) for i in 1:k],1))

π = s -> begin

if rand() < (1-β)^(k-1)

return πE(s)

else

return rand(Categorical(πθ(θs[rand(Pπ)], s)))

end

end

end

Ps = normalize([(1-β)^(i-1) for i in 1:k_max],1)

return Ps, θs

end

Algorithm 18.3. The SMILe al-
gorithm for training a stochas-
tic parameterized policy from ex-
pert demonstrations for an MDP 𝒫.
SMILe successively mixes in new
component policies with smaller
and smaller weight, while simulta-
neously reducing the probability of
acting according to the expert pol-
icy. The method returns the prob-
abilities Ps and parameterizations
θs for the component policies.

18.4. maximum margin inverse reinforcement learning 355

Consider using SMILe to train a policy on the mountain car problem where
the reward is not observed. We use the same features that were used for
dataset aggregation in example 18.3. Both dataset aggregation and SMILe
receive a new expert-labeled dataset with each iteration. Instead of accu-
mulating a larger dataset of expert-labeled data, SMILe trains a new policy
component using only the most recent data, mixing the new policy compo-
nent with the previous policy components.

Example 18.4. An example of us-
ing SMILe to learn a policy for
the mountain car problem. In con-
trast with dataset aggregation in
example 18.3, SMILe mixes the ex-
pert into the policy during rollouts.
This expert component, whose in-
fluence wanes with each iteration,
causes the initial rollouts to better
progress toward the goal.

356 chapter 18. imitation learning

An important part of this algorithm involves reasoning about the expected
return under a policy π for a weightingφ and initial state distribution b:

E
s∼b

[U(s)] = Eτ

[
d

∑
k=1

γk−1Rφ(s(k), a(k))

]
(18.2)

= Eτ

[
d

∑
k=1

γk−1φ>β(s(k), a(k))

]
(18.3)

= φ>
(

Eτ

[
d

∑
k=1

γk−1β(s(k), a(k))

])
(18.4)

= φ>µπ (18.5)

where τ corresponds to trajectories generated by π to depth d. Here, we intro-
duce the feature expectations µπ , which is the expected discounted accumulated
feature values. These feature expectations can be estimated from m rollouts, as
implemented in algorithm 18.4.

struct InverseReinforcementLearning

𝒫 # problem

b # initial state distribution

d # depth

m # number of samples

π # parameterized policy

β # binary feature mapping

μE # expert feature expectations

RL # reinforcement learning method

ϵ # tolerance

end

function feature_expectations(M::InverseReinforcementLearning, π)

𝒫, b, m, d, β, γ = M.𝒫, M.b, M.m, M.d, M.β, M.𝒫.γ

μ(τ) = sum(γ^(k-1)*β(s, a) for (k,(s,a)) in enumerate(τ))

τs = [simulate(𝒫, rand(b), π, d) for i in 1:m]

return mean(μ(τ) for τ in τs)

end

Algorithm 18.4. A structure for in-
verse reinforcement learning and
a method for estimating a feature
expectations vector from rollouts.

We can use the expert demonstrations to estimate the expert feature expecta-
tions µE, and we want to find a policy that matches these feature expectations as
closely as possible. At the first iteration, we begin with a randomized policy π(1)

and estimate its feature expectations that we denote µ(1). At iteration k, we find a
new φ(k), corresponding to a reward function R

φ(k)(s, a) = φ(k)>β(s, a), such

that the expert outperforms all previously found policies by the greatestmargin t:

maximize
t,φ

t

subject to φ>µE ≥ φ>µ(i) + t for i = 1, . . . , k− 1

‖φ‖2 ≤ 1

(18.6)

Equation (18.6) is a quadratic program that can be easily solved. We then solve
for a new policy π(k) using the reward function R(s, a) = φ(k)>β(s, a) and pro-
duce a new vector of feature expectations. Figure 18.2 illustrates this margin
maximization process.

We iterate until the margin is sufficiently small with t ≤ ε. At convergence, we
can solve for a mixed policy that attempts to have feature expectations as close as
possible to that of the expert policy:

minimize
λ

‖µE − µλ‖2

subject to λ ≥ 0

‖λ‖1 = 1

(18.7)

where µλ = ∑i λiµ
(i). The mixture weights λ combine the policies found at

each iteration. With probability λi, we follow policy π(i). Inverse reinforcement
learning is implemented in algorithm 18.5.

18.5 Maximum Entropy Inverse Reinforcement Learning

The inverse reinforcement learning approach from the previous section is un-
derspecified, meaning there are often multiple policies that can produce the
same feature expectations as the expert demonstrations. This section introduces
maximum entropy inverse reinforcement learning, which avoids this ambiguity by
preferring the policy that results in the distribution over trajectories that has
maximum entropy (appendix A.8).8 The problem can be transformed into one of 8 B.D. Ziebart, A. Maas, J. A. Bag-

nell, and A.K. Dey, ‘‘Maximum En-
tropy Inverse Reinforcement Learn-
ing,’’ in AAAI Conference on Artifi-
cial Intelligence (AAAI), 2008.

finding the best reward function parametersφ in a maximum likelihood estima-
tion problem given the expert data D.

Figure 18.2. A geometric visualiza-
tion of three example iterations of
the maximum-margin inverse rein-
forcement learning algorithm, go-
ing top to bottom. In each iteration,
the newweight vector points in the
direction perpendicular to the hy-
perplane that separates the expert
feature expectation vector from the
those of the previous policy with
the largest possible margin. The
margin decreases with each itera-
tion.

function calc_weighting(M::InverseReinforcementLearning, μs)

μE = M.μE

k = length(μE)

model = Model(Ipopt.Optimizer)

@variable(model, t)

@variable(model, ϕ[1:k] ≥ 0)

@objective(model, Max, t)

for μ in μs

@constraint(model, ϕ⋅μE ≥ ϕ⋅μ + t)

end

@constraint(model, ϕ⋅ϕ ≤ 1)

optimize!(model)

return (value(t), value.(ϕ))

end

function calc_policy_mixture(M::InverseReinforcementLearning, μs)

μE = M.μE

k = length(μs)

model = Model(Ipopt.Optimizer)

@variable(model, λ[1:k] ≥ 0)

@objective(model, Min, (μE - sum(λ[i]*μs[i] for i in 1:k))⋅

(μE - sum(λ[i]*μs[i] for i in 1:k)))

@constraint(model, sum(λ) == 1)

optimize!(model)

return value.(λ)

end

function optimize(M::InverseReinforcementLearning, θ)

π, ϵ, RL = M.π, M.ϵ, M.RL

θs = [θ]

μs = [feature_expectations(M, s->π(θ,s))]

while true

t, ϕ = calc_weighting(M, μs)

if t ≤ ϵ

break

end

copyto!(RL.ϕ, ϕ) # R(s,a) = ϕ⋅β(s,a)

θ = optimize(RL, π, θ)

push!(θs, θ)

push!(μs, feature_expectations(M, s->π(θ,s)))

end

λ = calc_policy_mixture(M, μs)

return λ, θs

end

Algorithm 18.5. Maximum
margin inverse reinforcement
learning, which computes a
mixed policy whose feature
expectations match those of given
expert demonstrations. We use
JuMP.jl to solve our constrained
optimization problems. This
implementation requires that the
provided reinforcement learning
struct has a weight vector ϕ that
can be updated with new values.
The method returns the stochastic
weightings λ and parameteri-
zations θs for the component
policies.

Any policy π induces a distribution over trajectories9 Pπ(τ). Different policies
9 For simplicity, this section as-
sumes a finite horizon and that the
state and action spaces are discrete,
making Pφ(τ) a probability mass.
To extend maximum entropy in-
verse reinforcement learning both
to problems with continuous state
and action spaces where the dy-
namics may be unknown, consider
guided cost learning: C. Finn, S.
Levine, and P. Abbeel, ‘‘Guided
Cost Learning: Deep Inverse Op-
timal Control via Policy Optimiza-
tion,’’ in International Conference on
Machine Learning (ICML), 2016.

produce different trajectory distributions. We are free to choose any of these
distributions over trajectories that match the expert feature expectations. The
principle of maximum entropy chooses the least informative distribution, which
corresponds to the one with maximum entropy.10 It can be shown that the least

10 For an introduction to this prin-
ciple, see E. T. Jaynes, ‘‘Informa-
tion Theory and StatisticalMechan-
ics,’’ Physical Review, vol. 106, no. 4,
pp. 620–630, 1957.

informative trajectory distribution has the form:

Pφ(τ) =
1

Z(φ)
exp(Rφ(τ)) (18.8)

where Pφ(τ) is the likelihood of a trajectory τ given reward parameterφ, and

Rφ(τ) =
d

∑
k=1

γk−1Rφ(s(k), a(k)) (18.9)

is the discounted trajectory reward. We make no assumption on the parameteri-
zation of Rφ(s(k), a(k)) other than that it is differentiable, allowing for representa-
tions such as neural networks. The normalization scalar Z(φ) ensures that the
probabilities sum to 1:

Z(φ) = ∑
τ

exp(Rφ(τ)) (18.10)

The summation is over all possible trajectories.
We seek to maximize the likelihood of the expert demonstrations under this

distribution:
max
φ

f (φ) = max
φ

∑
τ∈D

log Pφ(τ) (18.11)

We can rewrite the objective function f (φ) from equation (18.11):

f (φ) = ∑
τ∈D

log
1

Z(φ)
exp(Rφ(τ)) (18.12)

=

(
∑

τ∈D
Rφ(τ)

)
− |D| log Z(φ) (18.13)

=

(
∑

τ∈D
Rφ(τ)

)
− |D| log ∑

τ

exp(Rφ(τ)) (18.14)

We can attempt to optimize this objective function through gradient ascent.
The gradient of f is:

∇φ f =

(
∑

τ∈D
∇φRφ(τ)

)
− |D|

∑τ exp(Rφ(τ)) ∑
τ

exp(Rφ(τ))∇φRφ(τ)

(18.15)

=

(
∑

τ∈D
∇φRφ(τ)

)
− |D|∑

τ

Pφ(τ)∇φRφ(τ) (18.16)

=

(
∑

τ∈D
∇φRφ(τ)

)
− |D|∑

s
bγ,φ(s)∑

a
πφ(a | s)∇φRφ(s, a) (18.17)

If the reward function is linear with Rφ(s, a) = φ>β(s, a) as in the previous
section, then ∇φRφ(s, a) is simply β(s, a).

Updating the parameter vector φ thus requires both the discounted state
visitation frequency bγ,φ and the optimal policy under the current parameter
vector, πφ(a | s). We can obtain the optimal policy by running reinforcement
learning. To compute the discounted state visitation frequencies we can use
rollouts or take a dynamic programming approach.

If we take a dynamic programming approach to compute the discounted state
visitation frequencies, we can start with the initial state distribution b(1)γφ = b(s)
and iteratively work forward in time:

b(k+1)
γ,φ (s) = γ ∑

a
∑
s′

b(k)γ,φ(s)π(a | s)T(s′ | s, a) (18.18)

This version of maximum entropy inverse reinforcement learning is implemented
in algorithm 18.6.

18.6 Generative Adversarial Imitation Learning

In generative adversarial imitation learning (GAIL),11 we optimize a differentiable 11 J. Ho and S. Ermon, ‘‘Generative
Adversarial Imitation Learning,’’
in Advances in Neural Information
Processing Systems (NIPS), 2016.

parameterized policy πθ, often represented by a neural network. Rather than
provide a reward function, we use adversarial learning (appendix D.7). We also
train a discriminator Cφ(s, a), typically also a neural network, to return the prob-
ability it assigns to the state-action pair coming from the learned policy. The
process involves alternating between training this discriminator to become better

struct MaximumEntropyIRL

𝒫 # problem

b # initial state distribution

d # depth

π # parameterized policy π(θ,s)

Pπ # parameterized policy likelihood π(θ, a, s)

∇R # reward function gradient

RL # reinforcement learning method

α # step size

k_max # number of iterations

end

function discounted_state_visitations(M::MaximumEntropyIRL, θ)

𝒫, b, d, Pπ = M.𝒫, M.b, M.d, M.Pπ

𝒮, 𝒜, T, γ = 𝒫.𝒮, 𝒫.𝒜, 𝒫.T, 𝒫.γ

b_sk = zeros(length(𝒫.𝒮), d)

b_sk[:,1] = [pdf(b, s) for s in 𝒮]

for k in 2:d

for (si′, s′) in enumerate(𝒮)

b_sk[si′,k] = γ*sum(sum(b_sk[si,k-1]*Pπ(θ, a, s)*T(s, a, s′)

for (si,s) in enumerate(𝒮))

for a in 𝒜)

end

end

return normalize!(vec(mean(b_sk, dims=2)),1)

end

function optimize(M::MaximumEntropyIRL, D, ϕ, θ)

𝒫, π, Pπ, ∇R, RL, α, k_max = M.𝒫, M.π, M.Pπ, M.∇R, M.RL, M.α, M.k_max

𝒮, 𝒜, γ, nD = 𝒫.𝒮, 𝒫.𝒜, 𝒫.γ, length(D)

for k in 1:k_max

copyto!(RL.ϕ, ϕ) # update parameters

θ = optimize(RL, π, θ)

b = discounted_state_visitations(M, θ)

∇Rτ = τ -> sum(γ^(i-1)*∇R(ϕ,s,a) for (i,(s,a)) in enumerate(τ))

∇f = sum(∇Rτ(τ) for τ in D) - nD*sum(b[si]*sum(Pπ(θ,a,s)*∇R(ϕ,s,a)

for (ai,a) in enumerate(𝒜))

for (si, s) in enumerate(𝒮))

ϕ += α*∇f

end

return ϕ, θ

end

Algorithm 18.6. Maximum en-
tropy inverse reinforcement learn-
ing, which finds a stochastic pol-
icy that maximizes the likelihood
of the expert demonstrations un-
der a maximum-entropy trajectory
distribution. This implementation
computes the expected visitations
using dynamic programming over
all states, which requires that the
problem be discrete.

at distinguishing, and training the policy to look indistinguishable from the expert
demonstrations. The process is sketched in figure 18.3.

state

policy

π(a | s)

simulated state-action pairs expert state-action pairs

discriminator

P(simulated | s, a)

Figure 18.3. Instead of inferring a
reward function, generative adver-
sarial imitation learning optimizes
a discriminator to distinguish be-
tween simulated and expert state-
action pairs, and it optimizes a pol-
icy to appear indistinguishable to
the discriminator. The aim is to
eventually produce a policy that re-
sembles the expert.

The discriminator and policy have opposite objectives. Generative adversarial
imitation learning seeks to find a saddle-point (θ,φ) of the negative log loss of
the discriminator’s binary classification problem:12 12 The original paper also includes

an entropy term:

−λ E(s,a)∼D [− log πθ(a | s)]max
φ

min
θ

E(s,a)∼πθ

[
log(Cφ(s, a))

]
+ E(s,a)∼D

[
log(1− Cφ(s, a))

]
(18.19)

where we use (s, a) ∼ D to represent samples from the distribution represented
by the expert datasetD. We can alternate between gradient ascent onφ to increase
the objective and trust region policy optimization (section 12.4) on θ to reduce the
objective, generating the necessary trajectory samples from the policy to conduct
each of these steps. The discriminator provides a learning signal to the policy
similar to how a reward signal would if it were known.

18.7 Summary

• Imitation learning involves learning the desired behavior from expert demon-
stration without the use of a reward function.

• One type of imitation learning is behavioral cloning, which produces a stochas-
tic policy that maximizes the conditional likelihood of the actions in the dataset.

• When an expert can be queried multiple times, we can use iterative approaches
like dataset aggregation or stochastic mixing iterative learning.

• Inverse reinforcement learning involves inferring a reward function from expert
data and then using traditional methods for finding an optimal policy.

• Maximum margin inverse reinforcement learning attempts to find a policy that
matches the frequency of binary features found in the expert dataset.

• Maximum entropy inverse reinforcement learning frames the problem of find-
ing the best reward parameter as a maximum likelihood estimation problem,
which it tries to solve using gradient ascent.

• Generative adversarial imitation learning iteratively optimizes a discriminator
and a policy; the discriminator tries to discriminate between decisions made
by the policy and decisions made by the expert, and the policy attempts to
deceive the discriminator.

18.8 Exercises

Exercise 18.1. Consider applying behavioral cloning to a discrete problem where we have
been given expert demonstrations. An alternative could be to define a feature function
β(s) and represent the policy with a softmax distribution

π(a | s) ∝ exp(θ>a β(s))

We would then learn the parameters θa for each action from the expert data. Why might
we want to use this approach over one where we directly estimate a discrete distribution
for each state, with one parameter per state-action pair?

Solution: In imitation learning, we are generally limited to a relatively small set of expert
demonstrations. The distribution P(a | s) has (|A| − 1)|S| independent parameters that
must be learned, which is often prohibitively large. Expert demonstrations typically only
cover a small portion of the state-space. Even if P(a | s) can be reliably trained for the
states covered in the provided dataset, the resulting policy would be untrained in other
states. Using a feature function allows for generalization to unseen states.

Exercise 18.2. The section on behavioral cloning suggested using a maximum likelihood
approach for training a policy from expert data. This approach attempts to find the pa-
rameters of the policy that maximizes the likelihood assigned to the training examples.
In some problems, however, we know that assigning high probability to one incorrect
action is not as bad as assigning high probability to another incorrect action. For example,
predicting an acceleration of −1 in the mountain car problem when the expert dictates
an acceleration of 1 is worse than predicting an acceleration of 0. How might behavioral
cloning be modified to allow different penalties to be given to different misclassifications?

Solution: We can instead supply a cost function C(s, atrue, apred) that defines the cost of
predicting action apred for state s when the expert’s action is atrue. For example, with the
mountain-car problem we might use:

C(s, atrue, apred) = −|atrue − apred|

which penalizes greater deviations more than smaller deviations. The cost associated with
the expert’s action is typically zero.

If we have a stochastic policy π(a | s), then we seek to minimize the cost over our
dataset:

minimize
θ

∑
(s,atrue)∈D

∑
apred

C
(

s, atrue, apred
)

π
(

apred | s
)

This technique is called cost-sensitive classification.13 One benefit of cost-sensitive classifica- 13 C. Elkan, ‘‘The Foundations of
Cost-Sensitive Learning,’’ in Inter-
national Joint Conference on Artificial
Intelligence (IJCAI), 2001.

tion is that we can use a wide variety of off-the-shelf classification models like k-nearest
neighbors, support vector machines, or decision trees to train a policy.

Exercise 18.3. Provide an example of where maximum margin inverse reinforcement
learning does not uniquely define an optimal policy.

Solution: Maximum margin inverse reinforcement learning extracts binary features from
the expert data and seeks a reward function whose optimal policy produces trajectories
with the same frequencies of these binary features. There is no guarantee that multiple
policies do not produce the same feature expectations. For example, an autonomous car
that only makes left lane changes could have the same lane change frequencies as an
autonomous car that only makes right lane changes.

Exercise 18.4. Maximum margin inverse reinforcement learning measures how similar
a policy is to expert demonstrations using feature expectations. How is this similarity
measure affected if non-binary features are used?

Solution: If we use non-binary features, then it is possible that some features can get
larger than others, incentivizing the agent to match those features over those that tend
to be smaller. Scale is not the only issue. Even if all features are constrained to lie within
[0, 1], then a policy that consistently produces φ(s, a)1 = 0.5 will have the same feature
expectations as one that produces φ(s, a)1 = 0 half the time and φ(s, a)1 = 1 half the time.
Depending on what the feature encodes, this can result in very different policies. Any
set of continuous features can be discretized, and thus approximated by a set of binary
features.

Exercise 18.5. Suppose we are building a system for an elevator in a high-rise that has
to choose which floor to send an elevator. We have trained several policies to match the
feature expectations of expert demonstrations, such as how long a customer has to wait
for an elevator or how long they have to wait to get to their destination. We run multiple
rollouts for each policy and plot the relative duration spent on each floor. Which policy
should we prefer according to the principle of maximum entropy, assuming each policy
matches the feature expectations equally?

1 2 3 4 5 6 7 8 9 10
0

0.1

0.2

0.3

floor

re
la
tiv

e
du

ra
tio

n

Policy A

1 2 3 4 5 6 7 8 9 10

floor

Policy B

1 2 3 4 5 6 7 8 9 10

floor

Policy C

Solution: These distributions over relative duration are analogous to distributions over
trajectories for this elevator problem. In applying the principle of maximum entropy, we
prefer the distribution with most entropy. Hence, we would choose policy B, which in
being most uniform, has the greatest entropy.

Exercise 18.6. Consider the policy optimization step in generative adversarial imitation
learning. Rewrite the objective in the form of a reward function so that traditional rein-
forcement learning techniques can be applied.

Solution:We rewrite equation (18.19), dropping the terms dependent on the expert dataset,
and flip the sign to change from minimization over θ to a maximization over θ of the
reward, producing the surrogate reward function:

R̃φ(s, a) = − log Cφ(s, a)

Although R̃φ(s, a) may be quite different from the unknown true reward function, it can
be used to drive the learned policy into regions of the state-action space similar to those
covered by the expert.

Exercise 18.7. Explain how generative adversarial imitation learning could be changed
such that the discriminator takes in trajectories rather than state-action pairs. Why might
this be useful?

Solution: Changing generative adversarial imitation learning such that the discriminator
takes trajectories is straightforward, especially if the trajectories are of fixed length. The
expert dataset is split into trajectories, and the learned policy is used to produce trajectories,
just as it was before. Rather than operating on state-action pairs, the discriminator takes in
trajectories using a representation such as a recurrent neural network (appendix D.5) and
produces a classification probability. The objective function remains largely unchanged:

max
φ

min
θ

Eτ∼πθ

[
log(Cφ(τ))

]
+ Eτ∼D

[
log(1− Cφ(τ))

]
The advantage of running the discriminator over entire trajectories is that it can help

the discriminator capture features that are not apparent from individual state-action pairs,
which can result in better policies. For example, when looking at individual accelerations
and turn rates for an autonomous driving policy, there is very little for a discriminator to
learn. A discriminator trained to look at longer trajectories can see more of the vehicle’s
behavior, such as lane change aggressiveness and smoothness, to better match expert
driving demonstrations.14

14 This approach was used in A.
Kuefler, J. Morton, T.A. Wheeler,
and M. J. Kochenderfer, ‘‘Imitating
Driver Behavior with Generative
Adversarial Networks,’’ in IEEE In-
telligent Vehicles Symposium (IV),
2017.

